DTU Library

Computer programme for non-stationary moisture variations

Nielsen, Anker F.; Bach, Lars

Publication date:
1975

Document Version
Publisher's PDF, also known as Version of record

Link back to DTU Orbit

Citation (APA):
Nielsen, A. F., & Bach, L. (1975). *Computer programme for non-stationary moisture variations*. Technical University of Denmark, Department of Civil Engineering.

COMPUTER PROGRAMME FOR NON-STATIONARY MOISTURE VARIATIONS

by

Anker F. Nielsen
Thermal Insulation Laboratory

Lars Bach
Building Materials Laboratory

Technical University of Denmark

March 1975

Thermal Insulation Laboratory

Report no. 34

Introduction

Knowledge of the moisture content and distribution in wood products is important in connection with a number of wood engineering problems, e.g. drying of timber, creep of timber structures and other moisture-temperature-time-dependent processes in wood.

This paper gives a description of two computer programmes using the finite-difference method for computer-calculated moisture variations. The programmes solve the diffusion equation:

$$\frac{\partial MC}{\partial t} = \frac{\partial}{\partial x} \left(D \frac{\partial MC}{\partial x}\right)$$

where MC = moisture content (g/g or cm^3/cm^3 or %)
D = moisture diffusion coefficient (cm^2/sec)
t = time (sec)
x = place (cm)

The equation is solved with the finite-difference approximation using the Schmidt Method CRANK (2):

$$\Delta MC_n = \frac{D \cdot \Delta T}{\Delta x^2} \left(MC_{n+1} - MC_n + MC_{n-1}\right)$$

where Δx = thickness of layer (cm)
Δt = time difference (sec)
ΔMC_n = change of moisture content in Δt(-) of plane nr. n.
n = plane number (from 2 to n)

Both computer programmes have been written in FORTRAN IV, and have been run at the computing Center (NEUCC) at the Technical University of Denmark. The machine consists of an IBM 370/165 with a CALCOMP plotter. It could be necessary to rewrite the plotter part for use elsewhere.

This method will not always give a stable solution, as it is dependent of the factor $D \cdot \Delta t/\Delta x^2$, called the Fourier Number.

The solution is stable, if Fo is less than 0.1. The stability criterion can be found in KRISCHER (3), but it should be pointed out that his definition of the Fo-number is $x^2/D \cdot \Delta t$. The Krischer definition has been used in the computer programmes.

The moisture content in plane no. 1 is assumed in equilibrium with the surrounding air. It should be pointed out that it is necessary to make a sharp dissociation between the words plane and layer. The first plane no. 1 is the surface, and the last plane no. n is the center of the plate. The number of planes is equal to the number of layers + 1. It is assumed that the moisture variation takes place in a plate with the thickness 2a with the same surrounding media and climate on the two sides (the result would be the same for a plate with thickness "a" and the other side moisture-diffusion-tight).

Computer programme for cyclic conditions

This programme solves the diffusion equation for constant moisture diffusion coefficient and a cyclic moisture content at the surface. The variation could be sinusoidal or square wave.

For this programme it is convenient to use two non-dimensional variables:

$$\text{relative time} = RF = \frac{D \cdot t}{a^2}$$

$$\text{fractional moisture content} = FMC = \frac{MC(t) - MC(0)}{MC(max) - MC(0)}$$

where

- a = half thickness of wooden plate
- MC(t) = moisture content at time "t"
- MC(0) = initial moisture content
- MC(max) = maximum moisture content
- D = diffusion coefficient (constant)

In this way it is easy to use the obtained calculations for different dimensions, moisture content variations, time and diffusion coefficients.

The next pages is a list of the programme with comment cards, to make it understandable with a little knowledge of FORTRAN. The computing time, incl. plotter part, is approx. 7 sec for this example with REGION = 150 k. This programme has been used for the calculations in BACH & NIELSEN (1).

```
C       PROGRAM NAME    *** SINUS-DIFFUS ***
C       LARS BACH. LBM-DTH JULY 23,1973     (TEL.5245)
C       ANKER NIELSEN,LVI-DTH JULY 1973     (TEL.5352)
C
C       DIFFUSIONPROBLEM WITH VARIATING BOUNDARY CONDITION
C       SEE CRANK   MATHEMATICS OF DIFFUSION (FINITE DIFFERENCE METHOD)
C
C       XM(50)=MOISTURE CONTENT IN PERCENT
C       AM=MOISTURE CHANGE OF PLANE XM(K) TIME=ATIME+DT
C       XM(NLAG)=MOISTURE CONTENT OF CENTERPLANE
        DIMENSION  XM(60),QR(20),RXM(60)
C       NEXT 2 CARDS ONLY USED FOR PLOT
        DIMENSION RT(1000),RX(1000,11)
        DIMENSION RM(1000),RXX(1000),RYY(1000)
C
C
        READ(5,1)(QR(N),N=1,20)
C       QR=FIELD OF 80 COLOUMS FOR TEXT WRITING
      1 FORMAT(20A4)
C
        WRITE(6,2)(QR(N),N=1,20)
      2 FORMAT(1H1,20A4,///)
C
C
        READ(5,3) A,XMO,TID,DM,DIF,DT,TSTOP,NLAG,NUD
      3 FORMAT(7E10.1,I3,I3)
C
C       DT=TIME INTERVAL BETWEEN CALCULATIONS
C
C       A=HALF THICKNESS OF SAMPLE IN CENTIMETERS
C       XMO=MOISTURE CONTENT OF SURFACE PLANE
C       TID=TIME IN SECONDS FOR ONE PERIOD OF MOISTURE VARIATION
C       DM=AMPLITUDE OF MOISTURE VARIATION IN PERCENT
C       DIF=DIFFUSION COEFFICIENT
C       DT=TIMEINTERVAL IN SECONDS FOR EACH CALCULATION
C       TSTOP=MAX.VALUE OF TIME FOR WHICH THE PROGRAM WILL WORK
C        NLAG=NUMBER OF  PLANES WHERE MC IS CALCULATED (NLAG-1=NUMBER OF LAYERS)
C       NUD=NUMBER OF CALCULATIONS DONE FOR EACH ONE LISTED.
C       NZZ=NUMBER OG PLANES BETWEEN EACH LISTED IN OUTPUT
C       FO=FOURIER NUMBER (GREATER THAN 9.4 NECESSARY FOR STABILLITY)
C
C       DX=THICKNESS OF EACH LAYER IN CALCULATION
        NZZ=1
C
C       NEXT 4 CARDS ONLY USED FOR PLOT
        DT=TID/10000.
        XNUD=FLOAT(NUD)
        XY=TSTOP/(DT*XNUD)
        IF(XY.GT.1000.) DT=0.00001
C
        NN=NLAG-1
        DX=A/(FLOAT(NN))
        FO=(DX*DX)/(DIF*DT)
C       FO= SEE TROCKNUNGSTECHNIK I - O.KRISCHER + K.KROL (SPRINGER 63)
        TO=2.*3.1415/TID
C       TO=OMEGA (2*PHI/PERIOD OF OSCILATION IN SECONDS)
        AQ=DT*DIF/DX**2
C       AQ=SEE CRANK MATH. OF DIFFUSION(FINITE DIFFERENCE METHOD)
```

```
      N=0
      ATIME=0.0
      NX=0
C
C
      WRITE(6,300)FO
  300 FORMAT(1H0,10X,F11.7,'=FOURIER TAL LESS 9.4 NO STABILITY')
      WRITE(6,4)A,XMO,TID,DM,DIF,DT
    4 FORMAT(1H0,10X,F11.7,'=A=HALF THICKNESS OF SAMPLE IN CM',/,
     210X,F12.7, '=XMO=INITIAL MOISTURE CONTENT IN PERCT. (ALL)',/,
     310X,F12.7,'=TID=TIME IN SECONDS FOR ONE PERIOD OF MOI. VAR.',/,
     49X,0PF12.7,'=DM=AMPLITUDE OF MC-VARIATION',/,
     510X,F12.7,  '=DIF=DIFFUSION COEFFICIENT IN CM**2/SEC'/,
     610X,1PE12.6,'=DT=TIME INTERVAL (SEC) BETWEEN CALCULATIONS')
C
      WRITE(6,44) DX,AQ,TSTOP,NLAG,NUD
   44 FORMAT(1H ,
C
     *10X,1PE12.2,'=DX=THICKNESS OF EACH LAYER IN CM',/,
     *10X,1PE12.2,'=AQ=DT*DIF/DX**2 (SEE CRANK MATH. OF DIFFUS.)',/,
     *10X,1PE13.2,'=TSTOP=MAX.VALUE OF TIME FOR PROGRAM IN SEC',/,
     *10X,I12,     '=NLAG=(NO OF SURFACES IN CALCUL. WITH *NLAG-1* LAG',/
     *,9X,I12,     '=NUD=NUMBER OF CALCUL BETWEEN EACH RESULT LISTED'//)
      WRITE(6,30)
   30 FORMAT(1H0,6X,  'REL.TIME AVG.-MC RAND.=MC(1)  MC(2)  MC(3)  MC(4)
     *  MC(5)  MC(6)  MC(7)  MC(8)  MC(9) MC(10) MC(11)=MIDPLA',/)
C
      DO 5 K=1,NLAG
      XM(K)=XMO
      RM(K)=0.0
C
C     XM(K)=10.
C     RM(K)=-1.0
C
    5 CONTINUE
      RTIME=DIF*ATIME/(A*A)
      XMID=XMO
      RMID=0.
C
C
C
C
C     XM(1)=MOISTURE CONTENT SURFACE PLANE
C     XM(2) TO XM(NLAG-1)= MC OF PLANES BETWEEN SURFACE AND CENTER
C     XM(NLAG)= MOISTURE CONTENT  OF CENTER PLANE
C
      WRITE(6,59) ATIME,XMID,(XM(K),K=1,NLAG,NZZ)
   59 FORMAT(1H0,2X,F12.6,F8.3,5X,11F7.3,///,10X,'FRACTIONAL LISTING',/)
      WRITE(6,60) RTIME,RMID,(RM(K),K=1,NLAG,NZZ)
   60 FORMAT(1H0,2X,F12.6,F8.3,5X,11F7.3)
C
      TAL=0.
C     ZZZZZZZZZZZZZZZZZZZZZZ
      IF(FO.LE.9.4) GO TO 888
C     ZZZZZZZZZZZZZZZZZZZZZZ
    8 CONTINUE
      TAL=TAL+1.
      ATIME=DT*TAL
C     ACTUAL TIME IN SECONDS
C
```

```
      RTIME=DIF*ATIME/(A*A)
C     RELATIVE TIME = DIFFUSIONCOEF.*TIME/ HALF THICKNESS **2
C
C     SINUS VARIATION
      XM(1)=XM0+DM*SIN(TO*ATIME)
C
C     SQARE WAVW (NEXT 2 CARDS REMOVED IF SINUS VAR. WANTED)
      IF(XM(1).GE.15.)XM(1)=20.
      IF(XM(1).LT.15.)XM(1)=10.
C
C     NEXT CARD USED(NOT C IN COL. 1 ) IF STANDARD CHANGE IS WANTED
C     XM(1)=20.
C
      RXM(1)=(XM(1)-XM0)/DM
C
      DO 6 K=2,NN
      AM=AQ*(XM(K+1)-2.*XM(K)+XM(K-1))
      XM(K)=XM(K)+AM
      RXM(K)=(XM(K)-XM0)/DM
C
    6 CONTINUE
C
C
      BM=AQ*(XM(NN)-2.*XM(NLAG)+XM(NN))
      XM(NLAG)=XM(NLAG)+BM
C     XM(NLAG)=XM(NLAG)+AQ*(XM(NN)-XM(NLAG))*2.0
      RXM(NLAG)=(XM(NLAG)-XM0)/DM
C
C
      XSUM=0.0
      DO 11 I=2,NN
   11 XSUM=XSUM+XM(I)
      XMID=(0.5*XM(1)+0.5*XM(NLAG)+XSUM)/NN
C     XMID= AVERAGE MOISTURE CONTENT OF ALL LAYERS
C
      RXMI=(XMID-XM0)/DM
C     FRACTIONAL AVERAGE MOISTURECONTENT
C
C
      N=N+1
      IF(N.LT.NUD) GO TO 10
C     NUD=NUMBER OF CALCULATIONS DONE FOR EACH ONE LISTED.
C
C
C
C
C     NEXT CARDS ONLY USED FOR PLOT FROM HERE
      I=NX+1
      RT(I)=RTIME
      RM(I)=RXMI
      NX=I
C
      DO 999 K=1,NLAG
      RX(I,K)=RXM(K)
  999 CONTINUE
C
C
C
C  50 WRITE(6,51) RTIME,RXMI,(RXM(K),K=1,NLAG,NZZ),NX
   51 FORMAT(1H0,2X,F12.6,F8.3,5X,11F7.3,I6)
```

```
C
C      TO HERE
C
C
C      WRITE(6,7) ATIME,XMID,(XM(K),K=1,NLAG)
C    7 FORMAT(1H0,2X,E12.6,F8.2,5X,12F8.2)
C
       N=0
   10  IF(ATIME.GE.TSTOP) GO TO 100
       GO TO 8
  100  CONTINUE
C
C
C***********************************************************************
C***********************************************************************
C***********************************************************************
C
C      PLOTTING PROGRAM STARTS HERE
C
C
       CALL  PARALF(20.,20.)
       CALL  FAKTOR(0.5,0.5)
C      CALL  FAKTOR(0.5,0.5)
       CALL  PLIM(120)
       CALL  PLOT(-8.,24.,2)
       CALL  PLOT(51.,24.,1)
       CALL  PLOT(51.,-18.,1)
       CALL  PLOT(-8.,-18.,1)
       CALL  PLOT(-8.,24.,1)
C
C
       CALL  PLOT(-6.,22.,2)
       CALL  PLOT(49.,22.,1)
       CALL  PLOT(49.,-16.,1)
       CALL  PLOT(-6.,-16.,1)
       CALL  PLOT(-6.,22.,1)
       CALL  SYMBOL(38.,-15.,0.6,'LARS BACH SEP.1974',0.0,18)
C
C
       DO 101 I=1,NX
C
C
C*********************************************
       WRITE(6,55) RT(I),RM(I),(RX(I,K),K=1,NLAG,NZZ),NX,I
C      WRITE(6,55) RT(I),RM(I),(RX(I,K),K=1,NLAG,NZZ),NX,I
   55  FORMAT(1H0,2X,F12.6,F8.3,5X,11F7.3,2I6)
C*********************************************
C
  101  CONTINUE
       AMX=0.025
C      AMX=0.025
C      AMX=0.1
       YMA=10.
       AMY=0.1
       XMA=40.
C      XMA=40.
C      XMA=10.
       XLL=40.
C      AMX=MAALESTOK TILVAEKST X-AKSE
C      AMX=0.025 FOR X-AXCIS 1=40 CM
```

```
C     AMY=MAALESTOK TILVAEKST Y-AKSE
C     XMA=OMSAET-FAKTOR  X-VEKTOR OG BRUGER ENHED
C     XMA=40.  FOR X-AXCIS 1=40 CM
C     YMA=OMSAET-FAKTOR  Y-VEKTOR OG BRUGER ENHED
C     XLL=LENGTH X-AXCIS
C     DRAW X-AXCIS  ON 20,20 WITH 0,0  + TEKST
C     CALL AXE(0.0,0.0,XLL,0.,0.0,0.1,AMX,0.3,1,0,1,-1.,1.)
      CALL AXE(0.0,0.0,XLL,0.,0.0,0.1,AMX,0.3,1,0,1,-1.,1.)
C
C     DRAW Y-AXICIS ON 20,20 WITH 0,0  + TEKST
      CALL AXE(0.,-10.,20.,90.,-1.0,-0.9,AMY,0.4,1,1,1,1.,1.)
      CALL SYMBOL(-2.,-9.,0.8,    'FRACTIONAL MOISTURE CONTENT',90.,27)
C     CALL SYMBOL(+2.,+12.,1.0,'PERIOD OF SURFACE CHANGE =',0.0,26)
      CALL SYMBOL(+2.,+12.,1.0,'PERIOD OF SURFACE CHANGE =',0.0,26)
C     CALL SYMBOL(+2.,+12.,1.0,'STEP CHANGE OF ENVIRONMENT',0.0,26)
      CALL NUMBER(+25.,12.,1.0,TID,0.00,2)
C     CALL NUMBER(+25.,12.,1.0,TID,0.00,2)
      CALL SYMBOL(40.,0.5,0.6,     'RELATIVE TIME',0.0,13)
      CALL SYMBOL (41.,-0.5,0.6,'(D*T)/(A*A)',0.0,11)
      CALL SYMBOL(+2.,+14.,0.6,'FOURIER NUMBER =',0.0,16)
      CALL NUMBER(11.,14.,0.6,FO,0.00,1)
C
C
C     PLOTNING V.H.A. SUBROUTINE  * KURVE *
C
C     VI PLOTTER  ALLE  PUNKTER  VED AT GÅ TILBAGE
C     MOISTURE MATRIX  RX(I,K) HVOR TID ANGIVES VED VEKTOR RT(I)
C     OG  SNIT MOISTURE VED RM(I)
C     ANTAL FLADER ER ANGIVET*K=NLAG* OG ANTAL KURVE-TID PUNKTER *I=NX*
C     ANTAL LAG ER ANGIVET NLAG  OG ANTAL I'ER  I MATRIX NX
C
      KKK=240
C
      DO 103  K=1,NLAG,2
      DO 102  I=1,NX
C     RYY(I)=FRACTIONAL MC IN LAYER K  (-1.0 TO +1.0)
C     RM(I) =FRACTIONAL AVERAGE MC     (-1.0 TO +1.0)
C     RT(I) =FRACTIONAL TIME           ( 0.0 TO +4.0)
      RYY(I)=RX(I,K)
  102 CONTINUE
      NRSY=KKK+1
      KKK=NRSY
      CALL PLOT(0.0,0.0,2)
      CALL PLOT(RT(1),RYY(1),1)
      CALL KURVE(RT,RYY,NX,   1,XMA,YMA,0.0,0.,-1,NRSY,0.20)
  103 CONTINUE
C     CALL KURVE(RT, RM,NX,   1,XMA,YMA,0.0,0.,-1,NRSY,0.20)
C
C
      CALL PLTEND
  888 CONTINUE
C
      STOP
      END
```

Input data cards for the programme

1. card:	col 1-80	(10A4)	text writing
2. card:	col 1-10	(E10.1)	sample thickness (cm)
	col 11-20	(E10.1)	start moisture content
	col 21-30	(E10.1)	time for one period variation
	col 31-40	(E10.1)	amplitude for period variation
	col 41-50	(E10.1)	diffusion coefficient (cm^2/sec)
	col 51-60	(E10.1)	time steps between calculations
	col 61-70	(E10.1)	max. calculation time
	col 71-73	(I3)	number of planes in specimen
	col 74-76	(I3)	number of planes between each one listed

The printer output is seen on next page and consists of:

The input data.
The value of the Fourier-number. If it is less than 9.4 the calculation will not begin.
The moisture distribution, which is printed in fractional values of the moisture content versus the relative time.

The plotter output is shown for the same example.

MOISTURE CONTENT IN WOOD WITH SINUSODIAL VARIATING MC IN SURFACE LAYER

99.9999084=FOURIER TAL LESS 9.4 NO STABILITY

1.0000000=A=HALF THICKNESS OF SAMPLE IN CM
15.0000000=XMO=INITIAL MOISTURE CONTENT IN PERCT. (ALL)
1.0000000=TID=TIME IN SECONDS FOR ONE PERIOD OF MOI. VAR.
5.0000000=DM=AMPLITUDE OF MC-VARIATION
1.0000000=DIF=DIFFUSION COEFFICIENT IN CM**2/SEC
9.999999E-05=DT=TIME INTERVAL (SEC) BETWEEN CALCULATIONS
1.00E-01=DX=THICKNESS OF EACH LAYER IN CM
1.00E-02=AQ=DT*DIF/DX**2 (SEE CRANK MATH. OF DIFFUS.)
1.00E 00=TSTOP=MAX.VALUE OF TIME FOR PROGRAM IN SEC
11=NLAG=(NO OF SURFACES IN CALCUL. WITH *NLAG-1* LAG
100=NUD=NUMBER OF CALCUL BETWEEN EACH RESULT LISTED

REL.TIME	AVG.-MC	RAND.=MC(1)	MC(2)	MC(3)	MC(4)	MC(5)	MC(6)	MC(7)	MC(8)	MC(9)	MC(10)	MC(11)=MIDPLA		
0.0	15.000	15.000	15.000	15.000	15.000	15.000	15.000	15.000	15.000	15.000	15.000	15.000		

FRACTIONAL LISTING

0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0	0.0		
0.010000	0.121	1.000	0.478	0.171	0.047	0.011	0.002	0.000	0.000	0.000	0.0	0.0	100	1
0.020000	0.166	1.000	0.616	0.321	0.142	0.054	0.018	0.005	0.001	0.000	0.000	0.000	100	2
0.030000	0.201	1.000	0.683	0.416	0.225	0.109	0.047	0.018	0.007	0.002	0.001	0.000	100	3
0.040000	0.231	1.000	0.724	0.481	0.292	0.162	0.083	0.038	0.017	0.007	0.003	0.002	100	4
0.050000	0.257	1.000	0.752	0.528	0.345	0.210	0.119	0.062	0.031	0.014	0.007	0.005	100	5
0.060000	0.281	1.000	0.773	0.565	0.389	0.251	0.153	0.088	0.047	0.025	0.013	0.010	100	6
0.070000	0.303	1.000	0.789	0.594	0.424	0.288	0.185	0.113	0.066	0.037	0.023	0.018	100	7
0.080000	0.323	1.000	0.803	0.618	0.455	0.320	0.215	0.138	0.085	0.052	0.034	0.029	100	8
0.090000	0.343	1.000	0.814	0.638	0.481	0.348	0.242	0.162	0.105	0.068	0.048	0.041	100	9
0.100000	0.361	1.000	0.823	0.655	0.504	0.374	0.267	0.185	0.125	0.085	0.062	0.055	100	10
0.110000	0.378	1.000	0.831	0.671	0.524	0.396	0.291	0.207	0.145	0.103	0.079	0.071	100	11
0.120000	0.395	1.000	0.839	0.684	0.542	0.417	0.312	0.228	0.165	0.121	0.096	0.087	100	12
0.130000	0.411	1.000	0.845	0.696	0.558	0.436	0.333	0.249	0.185	0.140	0.113	0.105	100	13
0.140000	0.426	1.000	0.851	0.707	0.573	0.454	0.352	0.268	0.204	0.158	0.131	0.123	100	14
0.150000	0.441	1.000	0.856	0.716	0.587	0.470	0.370	0.287	0.223	0.177	0.150	0.141	100	15
0.160000	0.455	1.000	0.860	0.726	0.600	0.486	0.387	0.305	0.241	0.196	0.168	0.159	100	16
0.170000	0.469	1.000	0.865	0.734	0.611	0.500	0.404	0.323	0.259	0.214	0.187	0.178	100	17
0.180000	0.482	1.000	0.869	0.742	0.623	0.514	0.419	0.340	0.277	0.232	0.205	0.196	100	18
0.190000	0.495	1.000	0.873	0.749	0.633	0.527	0.434	0.356	0.294	0.250	0.223	0.214	100	19
0.200000	0.507	1.000	0.876	0.756	0.643	0.540	0.449	0.372	0.311	0.267	0.241	0.232	100	20
0.210000	0.519	1.000	0.880	0.763	0.653	0.552	0.463	0.388	0.328	0.285	0.258	0.250	100	21
0.220000	0.531	1.000	0.883	0.769	0.662	0.563	0.476	0.403	0.344	0.302	0.276	0.267	100	22
0.230000	0.543	1.000	0.886	0.775	0.670	0.574	0.489	0.417	0.360	0.318	0.293	0.284	100	23
0.240000	0.554	1.000	0.889	0.781	0.679	0.585	0.502	0.432	0.375	0.334	0.309	0.301	100	24
0.250000	0.565	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]	[illegible]


FRACTIONAL MOISTURE CONTENT
FOURIER NUMBER = 100.0
PERIOD OF SURFACE CHANGE = 1.00
RELATIVE TIME
(D*T)/(A*A)
LARS BACH SEP. 1974

<u>Computer programme for variable diffusion coefficient</u>

This programme solves the diffusion equation for variable diffusion coefficient and a known moisture content at the surface. For this programme it is not possible to use non-dimensional variables because of variable coefficient. The moisture content for each plane is read separately, so that the initial moisture content could be variable in the specimen.

This programme could be used for drying calculations, where the moisture content at the surface is in equilibrium with the surrounding air. The results could show the moisture distributions for different drying scedules as seen on the plots later.

The calculations in both programmes are carried out under the assumption that there is no temperature gradient, as this might influence the moisture transfer. This means rather complicated equations for heat and moisture:

moisture $$\frac{\partial MC}{\partial t} = \frac{\partial}{\partial x}\left(D \frac{\partial MC}{\partial x}\right) + \frac{\partial}{\partial x}\left(k \frac{\partial T}{\partial x}\right)$$

heat $$\frac{\partial T}{\partial t} = \frac{\partial}{\partial x}\left(a \frac{\partial T}{\partial x}\right)$$

where T = temperature
k = thermal moisture diffusivity
a = thermal diffusivity

In NIELSEN (5) a computer programme for the moisture equation is used on cellular concrete.

The computer programme in this paper is listed on the next pages. The computing time is approx. 3 sec for this example.

```
C       ANKER NIELSEN,THERMAL INSULATION LAB-DTH,MAR. 1975
C
C       DIFFUSIONPROBLEM WITH VARIATING BOUNDARY CONDITION
C       SEE CRANK   MATHEMATICS OF DIFFUSION (FINITE DIFFERENCE METHOD)
C
C       XM(50)=MOISTURE CONTENT IN PERCENT
C       AM=MOISTURE CHANGE OF PLANE XM(K) TIME=ATIME+DT
C       XM(NLAG)=MOISTURE CONTENT OF CENTERPLANE
        DIMENSION  XM(60),QR(20),ADIF(60)
C
C
        READ(5,1)(QR(N),N=1,20)
C       QR=FIELD OF 80 COLOUMS FOR TEXT WRITING
      1 FORMAT(20A4)
C
        WRITE(6,2)(QR(N),N=1,20)
      2 FORMAT(1H1,20A4,///)
C
C
        READ(5,3) A           ,DIF,DT,TSTOP,NLAG,NUD
      3 FORMAT(4E10.1,I3,I4)
C
C       DT=TIME INTERVAL BETWEEN CALCULATIONS
C
C       A=HALF THICKNESS OF SAMPLE IN CENTIMETERS
C       DIF=DIFFUSION COEFFICIENT
C       DT=TIMEINTERVAL IN SECONDS FOR EACH CALCULATION
C       TSTOP=MAX.VALUE OF TIME FOR WHICH THE PROGRAM WILL WORK
C        NLAG=NUMBER OF  PLANES WHERE MC IS CALCULATED (NLAG-1=NUMBER OF LAYERS)
C       NUD=NUMBER OF CALCULATIONS DONE FOR EACH ONE LISTED.
C       NZZ=NUMBER OG PLANES BETWEEN EACH LISTED IN OUTPUT
C       FO=FOURIER NUMBER (GREATER THAN 9.4 NECESSARY FOR STABILLITY)
C
C       DX=THICKNESS OF EACH LAYER IN CALCULATION
        NZZ=1
C
        XNUD=FLOAT(NUD)
C
        NN=NLAG-1
        DX=A/(FLOAT(NN))
        FO=(DX*DX)/(DIF*DT)
C       FO= SEE TROCKNUNGSTECHNIK I - O.KRISCHER + K.KROL (SPRINGER 63)
        AQ=DT*DIF/DX**2
C       AQ=SEE CRANK MATH. OF DIFFUSION(FINITE DIFFERENCE METHOD)
        N=0
        ATIME=0.0
C
C
        WRITE(6,300)FO
    300 FORMAT(1H0,10X,F11.2,'=FOURIER TAL LESS 9.4 NO STABILITY')
        WRITE(6,4)A,DIF,DT
      4 FORMAT(1H0,10X,F11.7,'=A=HALF THICKNESS OF SAMPLE IN CM',/,
       510X,1PE12.6,'=DIF=DIFFUSION COEFFICIENT IN CM**2/SEC'/,
       610X,1PE12.6,'=DT=TIME INTERVAL (SEC) BETWEEN CALCULATIONS')
C
        WRITE(6,44) DX,AQ,TSTOP,NLAG,NUD
     44 FORMAT(1H ,
C
```

```
     *10X,1PE12.2,'=DX=THICKNESS OF EACH LAYER IN CM',/,
     *10X,1PE12.2,'=AQ=DT*DIF/DX**2 (SEE CRANK MATH. OF DIFFUS.)',/,
     *10X,1PE13.2,'=TSTOP=MAX.VALUE OF TIME FOR PROGRAM IN SEC',/,
     *10X,I12,     '=NLAG=(NO OF SURFACES IN CALCUL. WITH *NLAG-1* LAG',/
     *,9X,I12,     '=NUD=NUMBER OF CALCUL BETWEEN EACH RESULT LISTED'//)
      WRITE(6,30)
   30 FORMAT(1H0,6X,  'ACT.TIME AVG.-MC RAND.=MC(1)  MC(2)  MC(3)  MC(4)
     *  MC(5)  MC(6)  MC(7)  MC(8)  MC(9) MC(10) MC(11)=MIDPLA',/)
C
      AQ=AQ/DIF
      READ(5,60) (XM(K),K=1,NLAG)
   60 FORMAT(8F8.3)
C
      XSUM=0.0
      DO 12 I=2,NN
   12 XSUM=XSUM+XM(I)
      XMID=(0.5*XM(1)+0.5*XM(NLAG)+XSUM)/NN
C
C
C
C     XM(1)=MOISTURE CONTENT SURFACE PLANE
C     XM(2) TO XM(NLAG-1)= MC OF PLANES BETWEEN SURFACE AND CENTER
C     XM(NLAG)= MOISTURE CONTENT  OF CENTER PLANE
C
      WRITE(6,59) ATIME,XMID,(XM(K),K=1,NLAG,NZZ)
   59 FORMAT(1H0,2X,F12.6,F8.3,5X,11F7.3,///,10X,'MOISTURE CONTENT ',/)
C
      TAL=0.
C     ZZZZZZZZZZZZZZZZZZZZZZZZ
      IF(FO.LE.9.4) GO TO 888
C     ZZZZZZZZZZZZZZZZZZZZZZZZ
    8 CONTINUE
      TAL=TAL+1.
      ATIME=DT*TAL
C     ACTUAL TIME IN SECONDS
C
C     DEF. OF SURFACE MOISTURE CONTENT TO BE INSERTED HEER
      XM(1)=30.-ATIME/21600.
      IF(XM(1).LT.10.) XM(1)=10.
C
      DO 62 I=2,NLAG
C     DEFI. OF DIFFUSION COEFFICIENT INSERTED HER
      IF(XM(I).GT.20.) GO TO 80
      ADIF(I)=(0.1+XM(I)*0.075)*1.0E-05
      GO TO 62
   80 ADIF(I)=(1.6+(XM(I)-20.)*0.783)*1.0E-05
   62 CONTINUE
C
C
C
      DO 6 K=2,NN
      AM=AQ*ADIF(K)*(XM(K+1)-2.*XM(K)+XM(K-1))
      XM(K)=XM(K)+AM
C
    6 CONTINUE
C
C
      BM=AQ*ADIF(NLAG)*(XM(NN)-2.*XM(NLAG)+XM(NN))
      XM(NLAG)=XM(NLAG)+BM
C
```

```
C
C
C
      N=N+1
      IF(N.LT.NUD) GO TO 10
C     NUD=NUMBER OF CALCULATIONS DONE FOR EACH ONE LISTED.
C
C
      XSUM=0.0
      DO 11 I=2,NN
   11 XSUM=XSUM+XM(I)
      XMID=(0.5*XM(1)+0.5*XM(NLAG)+XSUM)/NN
C     XMID= AVERAGE MOISTURE CONTENT OF ALL PLANES
C
      ITIME=IFIX(ATIME/360.)
      M1=IFIX(XM(1)*10.)
      M2=IFIX(XM(2)*10.)
      M3=IFIX(XM(3)*10.)
      M4=IFIX(XM(4)*10.)
      M5=IFIX(XM(5)*10.)
      WRITE(7,9001) ITIME,M1,M2,M3,M4,M5
 9001 FORMAT('90130',I6,15X,I4,4(7X,I4))
C
      WRITE(6,7) ATIME,XMID,(XM(K),K=1,NLAG)
    7 FORMAT(1H0,2X,E12.6,F8.2,5X,12F8.2)
C
      N=0
   10 IF(ATIME.GE.TSTOP) GO TO 100
      GO TO 8
  100 CONTINUE
  888 CONTINUE
      STOP
      END

ÅENTRY
```

Input data cards for the programme

1. card:	col 1-80	(20A4)	text writing
2. card:	col 1-10	(E10.1)	sample thickness (cm)
	col 11-20	(E10.1)	diffusion coefficient (cm^2/sec)
	col 21-30	(E10.1)	time step between calculations
	col 31-40	(E10.1)	max. calculation time
	col 41-43	(I3)	number of planes in specimen
	col 44-47	(I4)	number of planes between each one listed
3. card:	col 1-8	(F8.3)	moisture content plane 1 (surface)
	col 9-16	(F8.3)	- - - 2
	col 17-24	(F8.3)	- - - 3
	col 25-32	(F8.3)	- - - 4
	col 33-40	(F8.3)	- - - 5
	col 41-48	(F8.3)	- - - 6
	col 49-56	(F8.3)	- - - 7
	col 57-64	(F8.3)	- - - 8
4. card:	col 1-8	(F8.3)	- - - 9

as 3. card continued until the number of planes given in 2. card.

The printer output is seen on next page and consists of:

The input data

The value of the Fourier-number. If it is less than 9.4, the calculation will not begin.

The moisture distribution is printed versus the time.

69.44=FOURIER TAL LESS 9.4 NO STABILITY

5.0000000=A=HALF THICKNESS OF SAMPLE IN CM
1.000000E-05=DIF=DIFFUSION COEFFICIENT IN CM**2/SEC
3.600000E 02=DT=TIME INTERVAL (SEC) BETWEEN CALCULATIONS
5.00E-01=DX=THICKNESS OF EACH LAYER IN CM
1.44E-02=AQ=DT*DIF/DX**2 (SEE CRANK MATH. OF DIFFUS.)
1.00E 06=TSTOP=MAX.VALUE OF TIME FOR PROGRAM IN SEC
11=NLAG=(NO OF SURFACES IN CALCUL. WITH *NLAG-1* LAG
10=NUD=NUMBER OF CALCUL BETWEEN EACH RESULT LISTED

ACT.TIME	AVG.-MC	RAND.=MC(1)	MC(2)	MC(3)	MC(4)	MC(5)	MC(6)	MC(7)	MC(8)	MC(9)	MC(10)	MC(11)=MIDPLA
0.000000	30.000	30.000	30.000	30.000	30.000	30.000	30.000	30.000	30.000	30.000	30.000	30.000

MOISTURE CONTENT

0.360000E 04	29.98	29.83	29.93	29.97	29.99	30.00	30.00	30.00	30.00	30.00	30.00	30.00
0.720000E 04	29.95	29.67	29.82	29.91	29.96	29.98	29.99	30.00	30.00	30.00	30.00	30.00
0.108000E 05	29.91	29.50	29.71	29.83	29.91	29.95	29.98	29.99	30.00	30.00	30.00	30.00
0.144000E 05	29.87	29.33	29.58	29.74	29.85	29.91	29.95	29.98	29.99	29.99	30.00	30.00
0.180000E 05	29.81	29.17	29.45	29.65	29.78	29.87	29.92	29.95	29.97	29.99	29.99	29.99
0.216000E 05	29.76	29.00	29.32	29.54	29.70	29.81	29.88	29.93	29.96	29.97	29.98	29.99
0.252000E 05	29.70	28.83	29.18	29.43	29.62	29.75	29.84	29.90	29.93	29.96	29.97	29.97
0.288000E 05	29.63	28.67	29.04	29.32	29.53	29.68	29.79	29.86	29.91	29.94	29.95	29.96
0.324000E 05	29.56	28.50	28.90	29.21	29.44	29.61	29.73	29.82	29.88	29.91	29.93	29.94
0.360000E 05	29.49	28.33	28.76	29.09	29.34	29.53	29.67	29.77	29.84	29.88	29.91	29.91
0.396000E 05	29.42	28.17	28.62	28.98	29.25	29.45	29.61	29.72	29.80	29.85	29.88	29.89
0.432000E 05	29.34	28.00	28.48	28.86	29.15	29.37	29.54	29.66	29.75	29.81	29.84	29.85
0.468000E 05	29.26	27.83	28.34	28.73	29.05	29.29	29.47	29.61	29.70	29.77	29.81	29.82
0.504000E 05	29.17	27.67	28.19	28.61	28.94	29.20	29.40	29.54	29.65	29.72	29.76	29.77
0.540000E 05	29.09	27.50	28.05	28.49	28.83	29.11	29.32	29.48	29.59	29.67	29.72	29.73
0.576000E 05	29.00	27.33	27.90	28.36	28.73	29.02	29.24	29.41	29.53	29.62	29.67	29.68
0.612000E 05	28.91	27.17	27.76	28.24	28.62	28.92	29.16	29.34	29.47	29.56	29.61	29.63
0.648000E 05	28.81	27.00	27.61	28.11	28.51	28.82	29.07	29.26	29.41	29.50	29.56	29.57
0.684000E 05	28.72	26.83	27.47	27.98	28.39	28.73	28.99	29.19	29.34	29.44	29.50	29.51
0.720000E 05	28.62	26.67	27.32	27.85	28.28	28.63	28.90	29.11	29.26	29.37	29.43	29.45
0.756000E 05	28.52	26.50	27.17	27.72	28.17	28.52	28.81	29.03	29.19	29.30	29.37	29.39
0.792000E 05	28.42	26.33	27.03	27.59	28.05	28.42	28.72	28.94	29.11	29.23	29.30	29.32
0.828000E 05	28.32	26.17	26.88	27.46	27.93	28.32	28.62	28.86	29.04	29.16	29.23	29.25
0.864000E 05	28.22	26.00	26.73	27.33	27.82	28.21	28.53	28.77	28.96	29.08	29.15	29.18
0.900000E 05	28.11	25.83	26.58	27.20	27.70	28.11	28.43	28.68	28.87	29.00	29.08	29.10
0.936000E 05	28.01	25.67	26.43	27.06	27.58	28.00	28.33	28.59	28.79	28.92	29.00	29.02
0.972000E 05	27.90	25.50	26.29	26.93	27.46	27.89	28.23	28.50	28.70	28.84	28.92	28.95
0.100800E 06	27.79	25.33	26.14	26.80	27.34	27.78	28.13	28.41	28.62	28.76	28.84	28.87

The next two pages show the moisture distributions for two examples plotted with a computer programme from NIELSEN (5).

Example 1 is the one listed.

Example 2 has a change of the surface moisture content from 30% at the start to 12% at 40 hours and to 10% for times longer than 120 hours.

The diffusion coefficient is $0.1 \cdot 10^{-5}$ cm^2/sec at 0% and $1.6 \cdot 10^{-5}$ at 20% and $9.43 \cdot 10^{-5}$ at 30%. Between these points are straight lines.

Conclusion

These computer programmes are not very complicated, and it is hoped that they could be used as an aid for others to make own calculations. The greatest problem for these calculations is: Which values does the diffusion coefficient have? As the material is non-homogenous it must be expected that measurements will give different values in dependence of the wood structure.

The Thermal Insulation Laboratory has used gamma-ray-attenuation (NIELSEN (4)) for measurements of homogeneity and moisture distributions in cellular concrete (NIELSEN (5)) and bricks. Experiments of drying and infiltration of water in wood are going to start in 1975.



Fig. 2. Calculated moisture distribution in a sample of wood during drying from 30%. The diffusion coefficient is moisture dependent, and the surface moisture content is known.



Fig. 3. Calculated moisture distribution in a sample of wood during drying from 30%. The diffusion coefficient is moisture dependent, and the surface moisture content is known.

References

(1) BACH, L. og A. NIELSEN: A Numerical Method for Calculation of Moisture Content in Wood exposed to Cyclic Climatic Conditions. 1975 (in publication).

(2) CRANK, J: The Mathematics of Diffusion. Clarendon Press 1956.

(3) KRISCHER, O.: Die Wissenschaftlichen Grundlagen der Trocknungstechnik. Springer Verlag 1963.

(4) NIELSEN, A.F.: Gamma-Ray-Attenuation used for Measuring the Moisture Content and Homogeneity of Porous Concrete. Building Science, Vol. 7, pp 257-263, 1972.

(5) NIELSEN, A.F.: Moisture distributions in Cellular Concrete During Heat and Moisture Transfer. Licentiate Dissertation. Thermal Insulation Laboratory, Technical University of Denmark, 1974.